# $Id$
# Maintainer: Jan Alexander Steffens (heftig) <jan.steffens@gmail.com>
# Contributor: Jan de Groot <jgc@archlinux.org>

pkgname=anjuta-extras
pkgver=3.10.0
pkgrel=4
pkgdesc="Anjuta Integrated Development Environment extra plugins"
arch=('i686' 'x86_64')
license=('GPL')
depends=('anjuta')
makedepends=('intltool' 'itstool')
url="http://www.anjuta.org/"
source=(https://download.gnome.org/sources/$pkgname/${pkgver%.*}/$pkgname-$pkgver.tar.xz)
options=('!emptydirs')
sha256sums=('92f92e3c530f5649edfe4d8645e8cfd3161383cbc486c636ebb24cf2d2642213')

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr --sysconfdir=/etc --localstatedir=/var --disable-static
  sed -i -e 's/ -shared / -Wl,-O1,--as-needed\0/g' libtool
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
}
